
F07 – Linear Equations (LAPACK)

F07AUF (CGECON/ZGECON) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F07AUF (CGECON/ZGECON) estimates the condition number of a complex matrix A, where A has
been factorized by F07ARF (CGETRF/ZGETRF).

2 Specification

SUBROUTINE F07AUF(NORM, N, A, LDA, ANORM, RCOND, WORK, RWORK, INFO)
ENTRY cgecon(NORM, N, A, LDA, ANORM, RCOND, WORK, RWORK, INFO)
INTEGER N, LDA, INFO
real ANORM, RCOND, RWORK(∗)
complex A(LDA,∗), WORK(∗)
CHARACTER∗1 NORM

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine estimates the condition number of a complex matrix A, in either the 1-norm or the infinity-
norm:

κ1(A) = ‖A‖1‖A−1‖1 or κ∞(A) = ‖A‖∞‖A−1‖∞.

Note that κ∞(A) = κ1(A
H).

Because the condition number is infinite if A is singular, the routine actually returns an estimate of the
reciprocal of the condition number.

The routine should be preceded by a call to F06UAF to compute ‖A‖1 or ‖A‖∞, and a call to
F07ARF (CGETRF/ZGETRF) to compute the LU factorization of A. The routine then uses Higham’s
implementation of Hager’s method [1] to estimate ‖A−1‖1 or ‖A−1‖∞.

4 References

[1] Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with
applications to condition estimation ACM Trans. Math. Software 14 381–396

5 Parameters

1: NORM — CHARACTER*1 Input

On entry: indicates whether κ1(A) or κ∞(A) is estimated as follows:

if NORM = ’1’ or ’O’, then κ1(A) is estimated;
if NORM = ’I’, then κ∞(A) is estimated.

Constraint: NORM = ’1’, ’O’ or ’I’.

2: N — INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: A(LDA,∗) — complex array Input

Note: the second dimension of the array A must be at least max(1,N).

On entry: the LU factorization of A, as returned by F07ARF (CGETRF/ZGETRF).

4: LDA — INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F07AUF
(CGECON/ZGECON) is called.

Constraint: LDA ≥ max(1,N).

5: ANORM — real Input

On entry: if NORM = ’1’ or ’O’, the 1-norm of the original matrix A; if NORM = ’I’, the
infinity-norm of the original matrix A. ANORM may be computed by calling F06UAF with the
same value for the parameter NORM. ANORM must be computed either before calling F07ARF
(CGETRF/ZGETRF) or else from a copy of the original matrix A.

Constraint: ANORM ≥ 0.0.

6: RCOND — real Output

On exit: an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact
singularity is detected or the estimate underflows. If RCOND is less than machine precision,
then A is singular to working precision.

7: WORK(∗) — complex array Workspace

Note: the dimension of the array WORK must be at least max(1,2∗N).

8: RWORK(∗) — real array Workspace

Note: the dimension of the array RWORK must be at least max(1,2∗N).

9: INFO — INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = −i, the ith parameter had an illegal value. An explanatory message is output, and
execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ, and in practice is nearly always less
than 10ρ, although examples can be constructed where RCOND is much larger.

8 Further Comments

A call to this routine involves solving a number of systems of linear equations of the form Ax = b or
AHx = b; the number is usually 5 and never more than 11. Each solution involves approximately 8n2

real floating-point operations but takes considerably longer than a call to F07ASF (CGETRS/ZGETRS)
with 1 right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The real analogue of this routine is F07AGF (SGECON/DGECON).

9 Example

To estimate the condition number in the 1-norm of the matrix A, where

A =



−1.34 + 2.55i 0.28 + 3.17i −6.39− 2.20i 0.72− 0.92i
−0.17− 1.41i 3.31− 0.15i −0.15 + 1.34i 1.29 + 1.38i
−3.29− 2.39i −1.91 + 4.42i −0.14− 1.35i 1.72 + 1.35i
2.41 + 0.39i −0.56 + 1.47i −0.83− 0.69i −1.96 + 0.67i


 .

Here A is nonsymmetric and must first be factorized by F07ARF (CGETRF/ZGETRF). The true
condition number in the 1-norm is 231.86.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F07AUF Example Program Text
* Mark 15 Release. NAG Copyright 1991.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, LDA
PARAMETER (NMAX=8,LDA=NMAX)
CHARACTER NORM
PARAMETER (NORM=’1’)

* .. Local Scalars ..
real ANORM, RCOND
INTEGER I, INFO, J, N

* .. Local Arrays ..
complex A(LDA,NMAX), WORK(2*NMAX)
real RWORK(2*NMAX)
INTEGER IPIV(NMAX)

* .. External Functions ..
real F06UAF, X02AJF
EXTERNAL F06UAF, X02AJF

* .. External Subroutines ..
EXTERNAL cgecon, cgetrf

* .. Executable Statements ..
WRITE (NOUT,*) ’F07AUF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
* Read A from data file
*

READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*
* Compute norm of A
*

ANORM = F06UAF(NORM,N,N,A,LDA,RWORK)
*
* Factorize A
*

CALL cgetrf(N,N,A,LDA,IPIV,INFO)
*

WRITE (NOUT,*)
IF (INFO.EQ.0) THEN

*
* Estimate condition number
*

CALL cgecon(NORM,N,A,LDA,ANORM,RCOND,WORK,RWORK,INFO)
*

IF (RCOND.GE.X02AJF()) THEN
WRITE (NOUT,99999) ’Estimate of condition number =’,

+ 1.0e0/RCOND
ELSE

WRITE (NOUT,*) ’A is singular to working precision’
END IF

ELSE
WRITE (NOUT,*) ’The factor U is singular’

END IF
END IF
STOP

*
99999 FORMAT (1X,A,1P,e10.2)

END

9.2 Program Data

F07AUF Example Program Data
4 :Value of N

(-1.34, 2.55) ( 0.28, 3.17) (-6.39,-2.20) ( 0.72,-0.92)
(-0.17,-1.41) ( 3.31,-0.15) (-0.15, 1.34) ( 1.29, 1.38)
(-3.29,-2.39) (-1.91, 4.42) (-0.14,-1.35) ( 1.72, 1.35)
( 2.41, 0.39) (-0.56, 1.47) (-0.83,-0.69) (-1.96, 0.67) :End of matrix A

9.3 Program Results

F07AUF Example Program Results

Estimate of condition number = 1.50E+02

